

areg postestimation — Postestimation tools for areg

[Description](#) [Syntax for predict](#) [Menu for predict](#) [Options for predict](#)
[Remarks and examples](#) [References](#) [Also see](#)

Description

The following postestimation commands are available after `areg`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of estimates
<code>estat ic</code>	Akaike's and Schwarz's Bayesian information criteria (AIC and BIC)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>forecast</code> ¹	dynamic forecasts and simulations
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of coefficients
<code>linktest</code>	link test for model specification
<code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
<code>predict</code>	predictions, residuals, influence statistics, and other diagnostic measures
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of estimates
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

¹ `forecast` is not appropriate with `mi` estimation results.

Syntax for predict

```
predict [type] newvar [if] [in] [, statistic]
```

where $y_j = \mathbf{x}_j \mathbf{b} + d_{\text{absorbvar}} + e_j$ and *statistic* is

<i>statistic</i>	Description
Main	
xb	$\mathbf{x}_j \mathbf{b}$, fitted values; the default
stdp	standard error of the prediction
dresiduals	$d_{\text{absorbvar}} + e_j = y_j - \mathbf{x}_j \mathbf{b}$
* xbd	$\mathbf{x}_j \mathbf{b} + d_{\text{absorbvar}}$
* d	$d_{\text{absorbvar}}$
* residuals	residual
* score	score; equivalent to residuals

Unstarred statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample. Starred statistics are calculated only for the estimation sample, even when `if e(sample)` is not specified.

Menu for predict

Statistics > Postestimation > Predictions, residuals, etc.

Options for predict

Main

xb, the default, calculates the prediction of $\mathbf{x}_j \mathbf{b}$, the fitted values, by using the average effect of the absorbed variable. Also see **xbd** below.

stdp calculates the standard error of $\mathbf{x}_j \mathbf{b}$.

dresiduals calculates $y_j - \mathbf{x}_j \mathbf{b}$, which are the residuals plus the effect of the absorbed variable.

xbd calculates $\mathbf{x}_j \mathbf{b} + d_{\text{absorbvar}}$, which are the fitted values including the individual effects of the absorbed variable.

d calculates $d_{\text{absorbvar}}$, the individual coefficients for the absorbed variable.

residuals calculates the residuals, that is, $y_j - (\mathbf{x}_j \mathbf{b} + d_{\text{absorbvar}})$.

score is a synonym for **residuals**.

Remarks and examples

[stata.com](http://www.stata.com)

► Example 1

Continuing with [example 1](#) of [R] **areg**, we refit the model with robust standard errors and then obtain linear predictions and standard errors for those linear predictions.

```
. use http://www.stata-press.com/data/r13/auto2
(1978 Automobile Data)
```

```
. areg mpg weight gear_ratio, absorb(rep78) vce(robust)
(output omitted)
. predict xb_ar
(option xb assumed; fitted values)
. predict stdp_ar, stdp
```

We can obtain the same linear predictions by fitting the model with `xtreg, fe`, but we would first need to specify the panel structure by using `xtset`.

```
. xtset rep78
      panel variable:  rep78 (unbalanced)
. xtreg mpg weight gear_ratio, fe vce(robust)
(output omitted)
. predict xb_xt
(option xb assumed; fitted values)
. predict stdp_xt, stdp
. summarize xb_ar xb_xt stdp*
```

Variable	Obs	Mean	Std. Dev.	Min	Max
xb_ar	74	21.36805	4.286788	11.58643	28.07367
xb_xt	74	21.36805	4.286788	11.58643	28.07367
stdp_ar	74	.7105649	.1933936	.4270821	1.245179
stdp_xt	74	.8155919	.4826332	.0826999	1.709786

The predicted `xb` values above are the same for `areg` and `xtreg, fe`, but the standard errors for those linear predictions are different. The assumptions for these two estimators lead to different formulations for their standard errors. The robust variance estimates with `areg` are equivalent to the robust variance estimates using `regress`, including the panel dummies. The consistent robust variance estimates with `xtreg` are equivalent to those obtained by specifying `vce(cluster panelvar)` with that estimation command. For a theoretical discussion, see [Wooldridge \(2013\)](#), [Stock and Watson \(2008\)](#), and [Arellano \(2003\)](#); also see the [technical note](#) after example 3 of `[XT] xtreg`.

◀

▶ Example 2

We would like to use `linktest` to check whether the dependent variable for our model is correctly specified:

```
. use http://www.stata-press.com/data/r13/auto2, clear
(1978 Automobile Data)
. areg mpg weight gear_ratio, absorb(rep78)
(output omitted)
. linktest, absorb(rep78)
```

```
Linear regression, absorbing indicators          Number of obs   =          69
                                                F( 2,          62) =          46.50
                                                Prob > F         =          0.0000
                                                R-squared        =          0.6939
                                                Adj R-squared    =          0.6643
                                                Root MSE        =          3.3990
```

mpg	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
_hat	-.9305602	.9537856	-0.98	0.333	-2.83715 .9760302
_hatsq	.0462785	.0227219	2.04	0.046	.0008582 .0916989
_cons	19.24899	9.725618	1.98	0.052	-.1922457 38.69022
rep78	F(4, 62) =		1.278	0.288	(5 categories)

The squared residuals are significant in the regression for mpg on the linear and squared residuals; therefore, the test indicates that our dependent variable does not seem to be well specified. Let's transform the dependent variable into energy consumption, gallons per mile, fit the alternative model, and check the link test again.

```
. generate gpm = 1/mpg
. areg gpm weight gear_ratio, absorb(rep78)
  (output omitted)
. linktest, absorb(rep78)
```

Linear regression, absorbing indicators	Number of obs	=	69
	F(2, 62)	=	72.60
	Prob > F	=	0.0000
	R-squared	=	0.7436
	Adj R-squared	=	0.7187
	Root MSE	=	0.0068

gpm	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
_hat	.2842582	.7109124	0.40	0.691	-1.136835 1.705352
_hatsq	6.956965	6.862439	1.01	0.315	-6.760855 20.67478
_cons	.0175457	.0178251	0.98	0.329	-.0180862 .0531777
rep78	F(4, 62) =		0.065	0.992	(5 categories)

The link test supports the use of the transformed dependent variable.



References

Arellano, M. 2003. *Panel Data Econometrics*. Oxford: Oxford University Press.

Stock, J. H., and M. W. Watson. 2008. Heteroskedasticity-robust standard errors for fixed effects panel data regression. *Econometrica* 76: 155–174.

Wooldridge, J. M. 2013. *Introductory Econometrics: A Modern Approach*. 5th ed. Mason, OH: South-Western.

Also see

- [\[R\] areg](#) — Linear regression with a large dummy-variable set
- [\[U\] 20 Estimation and postestimation commands](#)